A Simple Introduction to Finite Element Analysis of Electromagnetic Problems

MATTHEW N. O. SADIKU

Abstract—**This paper is a tutorial introduction for an absolute beginner in finite element numerical analysis. The finite element method is applied to Laplacian electrostatic field problems. Suggestions are offered on how the basic concepts developed can be extended to finite element analysis of problem involving Poisson's or wave equation. A step-by-step procedure for coding the numerical method is presented; a useful, working Fortran program is also included.**

I. Introduction

THE finite element method has its origin in the field of structural analysis. Although the earlier mathematical treatment of the method was provided by Courant [1] in 1943, the method was not applied to electromagnetic (EM) problems until 1968. With the increasing application of the method in the numerical solution of continuum problems (of which EM problems are an integral part), an effort to apply the method in both undergraduate and graduate EM courses has gained momentum [2]–[4].

So far, little attempt has been made to present finite element method in simple terms which undergraduate students in electrical engineering can understand. It is disappointing to note that while a large amount of research papers and textbooks on the subject have been published by civil and mechanical engineers, it is hard for an electrical engineering student or a practicing electrical engineer, who is an absolute beginner in using finite element method, to get a simple, brief introductory material. Civil or mechanical engineering materials on finite element method are not easy for electrical engineers to understand. This is due to the fact that the terminology and notation used in those materials are usually inconsistent and unfamiliar to electrical engineers. It has been the general belief among electrical engineers that the method is not so easy for undergraduates to grasp. The students are merely asked to use an existing finite element program to generate solutions to specific problems assigned [3], [4]. This is justified on the ground that numerical methods constitute a fragment of an EM course and that the finite element method involves calculus of variation which the students may not be familiar with at the undergraduate level. Asking students to perform finite element analysis with either packaged commercial codes or faculty-developed programs can be dangerous if care is not taken. It is like asking students to use SPICE in an introductory course in networks. Although most commercial finite element packages are easy to use, yet pitfalls, not widely appreciated, exist. While, it is unreasonable to expect that students armed with only an introductory knowledge of finite element method write commercial codes, a basic understanding of the method and its limitations is vital.

This paper presents a tutorial introduction to the finite element method without requiring the reader to be familiar with variational calculus. This not only helps the curious user of an existing finite element program understand the mathematical basis of the method, it gives him the ability to modify the program if necessary or develop his own code. It is hoped that the paper will help the reader who is first being introduced to the finite element analysis.

Most EM problems involve either partial differential equations or integral equations. While partial differential equations are usually solved using the finite difference method or finite element method, integral equations are solved conveniently using moment method [5]. In contrast to other methods, the finite element method accounts for nonhomogeneity of the solution region. The systematic generality of the method makes it a versatile tool for a wide range of problems. As a result, flexible general-purpose computer programs can be constructed.

The finite element analysis of any problem involves basically four steps: 1) discretizing the solution region into finite number of subregions or elements, 2) deriving governing equations for a typical element, 3) assembling of all elements in the solution region, 4) and solving the system of equations obtained. Each of these steps will be discussed briefly in the subsequent sections.

II. Finite Element Discretization

We divide the solution region into a number of *finite elements* as illustrated in Fig. 1 where the region is subdivided into four nonoverlapping elements (two triangular and two quadrilateral) and seven nodes. We seek an approximation for the potential V_e within an element e and then interrelate the potential distributions in various elements such that the potential is continuous across inter-element boundaries. The approximate solution for the whole region is

$$V(x, y) \simeq \sum_{e=1}^{N} V_e(x, y) \tag{1}$$

Manuscript received January 18, 1989.
The author is with the Department of Electrical Engineering, Temple University, Philadelphia, PA 19122.
IEEE Log Number 8927100.



Fig. 1. A typical finite element subdivision of an irregular domain.

where N is the number of triangular elements into which the solution region is divided.

The most common form of approximation for V within an element is polynomial approximation, namely

$$V_e(x, y) = a + bx + cy \tag{2}$$

for triangular element and

$$V_e(x, y) = a + bx + cy + dxy \tag{3}$$

for quadrilateral element. The potential V_e in general is nonzero within element e but zero outside e. In view of the fact that quadrilateral elements are in general nonconforming elements (see Fig. 1), we prefer to use triangular elements throughout our analysis in this paper. Notice that our assumption of linear variation of potential within the triangular element as in (2) is the same as assuming that the electric field is uniform within the element, i.e.,

$$\vec{E}_e = -\nabla V_e = -(b\vec{a}_x + c\vec{a}_y). \tag{4}$$

III. Element Governing Equations

Consider a typical triangular element shown in Fig. 2. The potential V_{e1}, V_{e2}, and V_{e3} at nodes 1, 2, and 3, respectively, are obtained using (2), i.e.,

$$\begin{bmatrix} V_{e1} \\ V_{e2} \\ V_{e3} \end{bmatrix} = \begin{bmatrix} 1 & x_1 & y_1 \\ 1 & x_2 & y_2 \\ 1 & x_3 & y_3 \end{bmatrix} \begin{bmatrix} a \\ b \\ c \end{bmatrix}. \tag{5}$$

The coefficients a, b, and c are determined from (5) as

$$\begin{bmatrix} a \\ b \\ c \end{bmatrix} = \begin{bmatrix} 1 & x_1 & y_1 \\ 1 & x_2 & y_2 \\ 1 & x_3 & y_3 \end{bmatrix}^{-1} \begin{bmatrix} V_{e1} \\ V_{e2} \\ V_{e3} \end{bmatrix}. \tag{6}$$

Substituting this into (2) gives

$$V_e = [1 \quad x \quad y] \frac{1}{2A} \cdot \begin{bmatrix} (x_2y_3 - x_3y_2) & (x_3y_1 - x_1y_3) & (x_1y_2 - x_2y_1) \\ (y_2 - y_3) & (y_3 - y_1) & (y_1 - y_2) \\ (x_3 - x_2) & (x_1 - x_3) & (x_2 - x_1) \end{bmatrix} \cdot \begin{bmatrix} V_{e1} \\ V_{e2} \\ V_{e3} \end{bmatrix}$$



Fig. 2. Typical triangular element; the local node numbering 1-2-3 must proceed counterclockwise as indicated by the arrow.

or

$$V_e = \sum_{i=1}^{3} \alpha_i(x, y) V_{ei} \tag{7}$$

where

$$\alpha_1 = \frac{1}{2A}[(x_2y_3 - x_3y_2) + (y_2 - y_3)x + (x_3 - x_2)y], \tag{8a}$$

$$\alpha_2 = \frac{1}{2A}[(x_3y_1 - x_1y_3) + (y_3 - y_1)x + (x_1 - x_3)y], \tag{8b}$$

$$\alpha_3 = \frac{1}{2A}[(x_1y_2 - x_2y_1) + (y_1 - y_2)x + (x_2 - x_1)y], \tag{8c}$$

and A is the area of the element e, i.e.,

$$2A = \begin{vmatrix} 1 & x_1 & y_1 \\ 1 & x_2 & y_2 \\ 1 & x_3 & y_3 \end{vmatrix} = (x_1y_2 - x_2y_1) + (x_3y_1 - x_1y_3) + (x_2y_3 - x_3y_2)$$

or

$$A = \tfrac{1}{2}[(x_2 - x_1)(y_3 - y_1) - (x_3 - x_1)(y_2 - y_1)]. \tag{9}$$

The value of A is positive if the nodes are numbered counterclockwise (starting from any node) as shown by the arrow in Fig. 2. Note that (7) gives the potential at any point (x, y) within the element provided that the potentials at the vertices are known. This is unlike finite difference analysis where the potential is known at the grid points only. Also note that α_i are linear interpolation functions. They are called the *element shape functions* and they have the following properties [7]:

$$\alpha_i(x_j, y_j) = \begin{cases} 1 & i = j \\ 0 & i \neq j \end{cases} \tag{10a}$$

$$\sum_{i=1}^{3} \alpha_i(x, y) = 1. \tag{10b}$$

The shape functions α_1 and α_2, for example, are illustrated in Fig. 3.



(a) (b)

Fig. 3. Shape functions α_1 and α_2 for a triangular element.

The energy per unit length associated with the element e is given by [6], i.e.,

$$W_e = \tfrac{1}{2} \int \epsilon \left| \vec{E} \right|^2 dS = \tfrac{1}{2} \int \epsilon \left| \nabla V_e \right|^2 dS \qquad (11)$$

where a two-dimensional solution region free of charge ($\rho_s = 0$) is assumed. But from (7),

$$\nabla V_e = \sum_{i=1}^{3} V_{ei} \nabla \alpha_i . \qquad (12)$$

Substituting (12) into (11) gives

$$W_e = \tfrac{1}{2} \sum_{i=1}^{3} \sum_{j=1}^{3} \epsilon V_{ei} \left[\int \nabla \alpha_i \cdot \nabla \alpha_j \, dS \right] V_{ej} . \qquad (13)$$

If we define the term in brackets as

$$C_{ij}^{(e)} = \int \nabla \alpha_i \cdot \nabla \alpha_j \, dS, \qquad (14)$$

we may write (13) in matrix form as

$$W_e = \tfrac{1}{2} \epsilon [V_e]^T [C^{(e)}] [V_e] \qquad (15)$$

where the superscript T denotes the transpose of the matrix,

$$[V_e] = \begin{bmatrix} V_{e1} \\ V_{e2} \\ V_{e3} \end{bmatrix} \qquad (16)\text{a}$$

and

$$[C^{(e)}] = \begin{bmatrix} C_{11}^{(e)} & C_{12}^{(e)} & C_{13}^{(e)} \\ C_{21}^{(e)} & C_{22}^{(e)} & C_{23}^{(e)} \\ C_{31}^{(e)} & C_{32}^{(e)} & C_{33}^{(e)} \end{bmatrix} . \qquad (16)\text{b}$$

The matrix $[C^{(e)}]$ is usually called the *element coefficient matrix* (or "stiffness matrix" in structural analysis). The matrix element $C_{ij}^{(e)}$ of the coefficient matrix may be regarded as the coupling between nodes i and j; its value is obtained from (8) and (14). For example,

$$\begin{aligned} C_{12}^{(e)} &= \int \nabla \alpha_1 \cdot \nabla \alpha_2 \, dS \\ &= \frac{1}{4A^2} \left[(y_2 - y_3)(y_3 - y_1) \right. \\ &\quad \left. + (x_3 - x_2)(x_1 - x_3) \right] \int dS \\ &= \frac{1}{4A} \left[(y_2 - y_3)(y_3 - y_1) \right. \\ &\quad \left. + (x_3 - x_2)(x_1 - x_3) \right] . \end{aligned} \qquad (17)\text{a}$$

Similarly,

$$C_{13}^{(e)} = \frac{1}{4A} \left[(y_2 - y_3)(y_1 - y_2) + (x_3 - x_2)(x_2 - x_1) \right] \qquad (17)\text{b}$$

$$C_{23}^{(e)} = \frac{1}{4A} \left[(y_3 - y_1)(y_1 - y_2) + (x_1 - x_3)(x_2 - x_1) \right] \qquad (17)\text{c}$$

$$C_{11}^{(e)} = \frac{1}{4A} \left[(y_2 - y_3)^2 + (x_3 - x_2)^2 \right] \qquad (17)\text{d}$$

$$C_{22}^{(e)} = \frac{1}{4A} \left[(y_3 - y_1)^2 + (x_1 - x_3)^2 \right] \qquad (17)\text{e}$$

$$C_{33}^{(e)} = \frac{1}{4A} \left[(y_1 - y_2)^2 + (x_2 - x_1)^2 \right] . \qquad (17)\text{f}$$

Also

$$C_{21}^{(e)} = C_{12}^{(e)}, \; C_{31}^{(e)} = C_{13}^{(e)}, \; C_{32}^{(e)} = C_{23}^{(e)} . \qquad (18)$$

IV. Assembling of all Elements

Having considered a typical element, the next step is to assemble all such elements in the solution region. The energy associated with the assemblage of elements is

$$W = \sum_{e=1}^{N} W_e = \tfrac{1}{2} \epsilon [V]^T [C] [V] \qquad (19)$$

where

$$[V] = \begin{bmatrix} V_1 \\ V_2 \\ \vdots \\ V_n \end{bmatrix} , \qquad (20)$$

n is the number of nodes, N is the number of elements, and $[C]$ is called the over-all or *global coefficient matrix* which is the assemblage of individual element coefficient matrices.

The process by which individual element coefficient matrices are assembled to obtain the global coefficient matrix is best illustrated with an example. Consider the finite element mesh consisting of three finite elements as shown in Fig. 4. Observe the numberings of the mesh. The numbering of nodes as 1, 2, 3, 4, and 5 is called *global* numbering. The numbering *i-j-k* is called *local* numbering and it corresponds with 1-2-3 of the element in Fig. 2. For example, for element 3 in Fig. 4, the local numbering 3-5-4 corresponds with 1-2-3 of the element in Fig. 2. Note that the local numbering must be in counterclockwise sequence starting from any node of the element. For element 3, we could choose 4-3-5 instead of 3-5-4 to be correspond with 1-2-3 of the element in Fig. 2. Thus, the numbering in Fig. 4 is not unique. Assuming the particular numbering in Fig. 4, the global coefficient



Fig. 4. Assembly of three elements; *i-j-k* corresponds to local numbering (1-2-3) of the element in Fig. 2.

matrix is expected to have the form

$$[C] = \begin{bmatrix} C_{11} & C_{12} & C_{13} & C_{14} & C_{15} \\ C_{21} & C_{22} & C_{23} & C_{24} & C_{25} \\ C_{31} & C_{32} & C_{33} & C_{34} & C_{35} \\ C_{41} & C_{42} & C_{43} & C_{44} & C_{45} \\ C_{51} & C_{52} & C_{53} & C_{54} & C_{55} \end{bmatrix} \tag{21}$$

which is a 5 × 5 matrix since five nodes ($n = 5$) are involved. Again, C_{ij} is the coupling between nodes i and j. We obtain C_{ij} by using the fact that the potential distribution must be continuous across interelement boundaries. The contribution to the i, j position in $[C]$ comes from all elements containing nodes i and j. For example, in Fig. 4, elements 1 and 2 have node 1 in common; hence

$$C_{11} = C_{11}^{(1)} + C_{11}^{(2)}. \tag{22a}$$

Node 2 belongs to element 1 only; hence

$$C_{22} = C_{33}^{(1)}. \tag{22b}$$

Node 4 belongs to elements 1, 2, and 3; hence

$$C_{44} = C_{22}^{(1)} + C_{33}^{(2)} + C_{33}^{(3)}. \tag{22c}$$

Nodes 1 and 4 belong simultaneously to elements 1 and 2; hence

$$C_{14} = C_{41} = C_{12}^{(1)} + C_{13}^{(2)}. \tag{22d}$$

Since there is no coupling (or direct link) between nodes 2 and 3,

$$C_{23} = C_{32} = 0. \tag{22e}$$

Continuing in this manner, we obtain all the terms in the global coefficient matrix by inspection of Fig. 4 as

$$[C] = \begin{bmatrix} C_{11}^{(1)} + C_{11}^{(2)} & C_{13}^{(1)} & C_{12}^{(2)} & C_{12}^{(1)} + C_{13}^{(2)} & 0 \\ C_{31}^{(1)} & C_{33}^{(1)} & 0 & C_{32}^{(1)} & 0 \\ C_{21}^{(2)} & 0 & C_{22}^{(2)} + C_{11}^{(3)} & C_{23}^{(2)} + C_{13}^{(3)} & C_{12}^{(3)} \\ C_{21}^{(1)} + C_{31}^{(2)} & C_{23}^{(1)} & C_{32}^{(2)} + C_{31}^{(3)} & C_{22}^{(1)} + C_{33}^{(2)} + C_{33}^{(3)} & C_{32}^{(3)} \\ 0 & 0 & C_{21}^{(3)} & C_{23}^{(3)} & C_{22}^{(3)} \end{bmatrix}. \tag{23}$$

Note that element coefficient matrices overlap at nodes shared by elements and that there are 27 terms (9 for each of the 3 elements) in the global coefficient matrix $[C]$. Also note the following properties of the matrix $[C]$:

1) It is symmetric ($C_{ij} = C_{ji}$) just as the element coefficient matrix.

2) Since $C_{ij} = 0$ if no coupling exists between nodes i and j, it is evident that for a large number of elements $[C]$ becomes sparse and banded.

3) It is singular. Although this is not so obvious, it can be shown using the element coefficient matrix of (16)b. Adding columns 2 and 3 to column 1 yields zeros in column 1 in (16)b.

IV. Solving the Resulting Equations

It can be shown that Laplace's (or Poisson's) equation is satisfied when the total energy in the solution region is minimum. Thus, we require that the partial derivatives of W with respect to each nodal value of the potential be zero, i.e.,

$$\frac{\partial W}{\partial V_1} = \frac{\partial W}{\partial V_2} = \cdots\cdots\cdots = \frac{\partial W}{\partial V_n} = 0$$

or

$$\frac{\partial W}{\partial V_k} = 0 \qquad k = 1, 2, \cdots\cdots\cdots, n. \tag{24}$$

For example, to get $\partial W/\partial V_1 = 0$ for the finite element mesh of Fig. 4, we substitute (21) into (19) and take the partial derivative of W with respect to V_1. We obtain

$$0 = \frac{\partial W}{\partial V_1} = 2V_1C_{11} + V_2C_{12} + V_3C_{13} + V_4C_{14} + V_5C_{15} + V_2C_{21} + V_3C_{31} + V_4C_{41} + V_5C_{51}$$

or

$$0 = V_1C_{11} + V_2C_{12} + V_3C_{13} + V_4C_{14} + V_5C_{15}. \tag{25}$$

In general, $\partial W/\partial V_k = 0$ leads to

$$0 = \sum_{i=1}^{n} V_iC_{ik} \tag{26}$$

where n is the number of nodes in the mesh. By writing (26) for all nodes $k = 1, 2, \cdots\cdots, n$, we obtain a set of simultaneous equations from which the solution of $[V]^T = [V_1, V_2, \cdots\cdots, V_n]$ can be found. This can be done in two ways similar to those used in solving finite difference equations obtained from Laplace's (or Poisson's) equation [8].

A) Iteration Method: Suppose node 0 is connected to m nodes as shown in Fig. 5. Using the idea in (25),

$$0 = V_oC_{oo} + V_1C_{o1} + V_2C_{o2} + \cdots\cdots\cdots + V_mC_{om}$$



Fig. 5. Node 0 connected to m other nodes.

or

$$V_o = -\frac{1}{C_{oo}} \sum_{k=1}^{m} V_k C_{ok}. \tag{27}$$

Thus, if the potentials at nodes connected to 0 are known, we can determine V_o using (27). The iteration process begins by setting the potentials at the *free* nodes (where the potential is unknown) equal to zero or to the average potential [9]

$$V_{ave} = \tfrac{1}{2}(V_{min} + V_{max}) \tag{28}$$

where V_{min} and V_{max} are the minimum and maximum values of V at the *fixed* nodes (where the potential V is prescribed or known). With these initial values, the potentials at the free nodes are calculated using (27). At the end of the first iteration, when the new values have been calculated for all the free nodes, they become the old values for the second iteration. The procedure is repeated until the change between subsequent iterations is neglible enough.

B) Band Matrix Method: If all free nodes are numbered first and the fixed nodes last, (19) can be written such that

$$W = \tfrac{1}{2}\epsilon [V_f \quad V_p] \begin{bmatrix} C_{ff} & C_{fp} \\ C_{pf} & C_{pp} \end{bmatrix} \begin{bmatrix} V_f \\ V_p \end{bmatrix} \tag{29}$$

where subscripts f and p, respectively, refer to nodes with free and fixed (or prescribed) potentials. Since V_p is constant (it consists of known, fixed values), we only differentiate with respect to V_f so that applying (24) to (29) yields [7]

$$[C_{ff} \quad C_{fp}] \begin{bmatrix} V_f \\ V_p \end{bmatrix} = 0$$

or

$$[C_{ff}][V_f] = -[C_{fp}][V_p]. \tag{30}$$

This equation can be written as

$$[A][V] = [B] \tag{31a}$$

or

$$[V] = [A]^{-1}[B] \tag{31b}$$

where $[V] = [V_f]$, $[A] = [C_{ff}]$, $[B] = -[C_{fp}][V_p]$. Since $[A]$ is, in general, nonsingular, the potential at the free nodes can be found using (31). We can solve for $[V]$ in (31)a using Gaussian elimination technique. We can also solve for $[V]$ in (31)b using matrix inversion if the size of the matrix to be inverted is not large.

Notice that as from (11) onward, the solution has been restricted to two-dimensional problem involving Laplace's equation, $\nabla^2 V = 0$. The basic concepts developed in this paper can be extended to finite element analysis of problems involving Poisson's equation ($\nabla^2 V = -\rho_v/\epsilon$, $\nabla^2 \vec{A} = -\mu \vec{J}$) or wave equation ($\nabla^2 \Phi - \gamma^2 \Phi = 0$).

VI. Illustrative Examples

Implementation of the concepts discussed in the preceeding sections is illustrated with two examples. The first example is a necessary preparation for the second where the numerical process is developed into a Fortran program. The code developed is simple and self-explanatory since it is intended for instructional purposes. The notations used in the programs are as close as possible to those used in previous sections; some are defined wherever necessary.

Example 1: Consider the two-element mesh shown in Fig. 6(a). Using finite element method, determine the potentials within the mesh.

Solution: The element coefficient matrices can be calculated using (17) and (18). However, our calculations will be easier if we define

$$P_1 = (y_2 - y_3),\ P_2 = (y_3 - y_1),\ P_3 = (y_1 - y_2)$$
$$Q_1 = (x_3 - x_2),\ Q_2 = (x_1 - x_3),\ Q_3 = (x_2 - x_1). \tag{32}$$

With P_i and Q_i ($i = 1, 2, 3$ are the local node numbers), each term in the element coefficient matrix is found as

$$C_{ij}^{(e)} = \frac{1}{4A}[P_i P_j + Q_i Q_j] \tag{33}$$

where $A = \frac{1}{2}(P_2 Q_3 - P_3 Q_2)$. It is evident that (33) is more convenient to use that (17) and (18). For element 1 consisting of nodes 1-2-4 corresponding to the local numbering 1-2-3 as in Fig. 6(b),

$$P_1 = -1.3 \quad P_2 = 0.9 \quad P_3 = 0.4,$$
$$Q_1 = -0.2 \quad Q_2 = -0.4 \quad Q_3 = 0.6,$$
$$A = \tfrac{1}{2}(0.54 + 0.16) = 0.35.$$

Substituting all these into (33) gives

$$[C^{(1)}] = \begin{bmatrix} 1.236 & -0.7786 & -0.4571 \\ -0.7786 & 0.6929 & 0.0857 \\ -0.4571 & 0.0857 & 0.3714 \end{bmatrix}. \tag{34}$$



Fig. 6. For example 1: (a) two-element mesh, (b) local and global numbering of the elements.

Similarly, for element 2 consisting of nodes 2-3-4 corresponding to local numbering 1-2-3 as in Fig. 6(b),

$$P_1 = -0.6 \quad P_2 = 1.3 \quad P_3 = -0.7,$$

$$Q_1 = -0.9 \quad Q_2 = 0.2 \quad Q_3 = 0.7,$$

$$A = \tfrac{1}{2}(0.91 + 0.14) = 0.525.$$

Hence,

$$[C^{(2)}] = \begin{bmatrix} 0.5571 & -0.4571 & -0.1 \\ -0.4571 & 0.8238 & -0.3667 \\ -0.1 & -0.3667 & 0.4667 \end{bmatrix}. \quad (35)$$

Applying (30) gives

$$\begin{bmatrix} C_{22} & C_{24} \\ C_{42} & C_{44} \end{bmatrix} \begin{bmatrix} V_2 \\ V_4 \end{bmatrix} = -\begin{bmatrix} C_{21} & C_{23} \\ C_{41} & C_{43} \end{bmatrix} \begin{bmatrix} V_1 \\ V_3 \end{bmatrix}. \quad (36)$$

This can be written in a more convenient form as

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & C_{22} & 0 & C_{24} \\ 0 & 0 & 1 & 0 \\ 0 & C_{42} & 0 & C_{44} \end{bmatrix} \begin{bmatrix} V_1 \\ V_2 \\ V_3 \\ V_4 \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ -C_{21} & -C_{23} \\ 0 & 1 \\ -C_{41} & -C_{43} \end{bmatrix} \begin{bmatrix} V_1 \\ V_3 \end{bmatrix} \quad (37)\text{a}$$

or

$$[C][V] = [B]. \quad (37)\text{b}$$

The terms of the global coefficient matrix are obtained as follows:

$$C_{22} = C_{22}^{(1)} + C_{11}^{(2)} = 0.6929 + 0.5571 = 1.25$$

$$C_{24} = C_{23}^{(1)} + C_{13}^{(2)} = 0.0857 - 0.1 = -0.0143$$

$$C_{44} = C_{33}^{(1)} + C_{33}^{(2)} = 0.3714 + 0.4667 = 0.8381$$

$$C_{21} = C_{21}^{(1)} = -0.7786$$

$$C_{23} = C_{12}^{(2)} = -0.4571$$

$$C_{41} = C_{31}^{(1)} = -0.4571$$

$$C_{43} = C_{32}^{(2)} = -0.3667.$$

Note that we follow local numbering for the element coefficient matrix and global numbering for the global coefficient matrix. Thus, the square matrix $[C]$ is obtained as

$$[C] = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1.25 & 0 & -0.0143 \\ 0 & 0 & 1 & 0 \\ 0 & -0.0143 & 0 & 0.8381 \end{bmatrix} \quad (38)$$

and the matrix $[B]$ on the right-hand side of (37)a is obtained as

$$[B] = \begin{bmatrix} 0 \\ 4.571 \\ 10.0 \\ 3.667 \end{bmatrix}. \quad (39)$$

By inverting matrix $[C]$ in (38), we obtain

$$[V] = [C]^{-1}[B] = \begin{bmatrix} 0 \\ 3.708 \\ 10.0 \\ 4.438 \end{bmatrix}.$$

Thus, $V_1 = 0$, $V_2 = 3.708$, $V_3 = 10$, and $V_4 = 4.438$. Once the values of the potentials at the nodes are known, the potential at any point within the mesh can be determined using (7).

Example 2: Write a Fortran program to solve the Laplace's equation using finite element method. Apply the program to the two-dimensional problem shown in Fig. 7(a).

Solution: The solution region is divided into 25 three-node triangular elements with total number of nodes being 21 as shown in Fig. 7(b). This is a necessary step in order to have input data defining the geometry of the problem. Based on the discussions in Sections II–IV, a general Fortran program for solving problems involving Laplace's equation using three-node triangular elements is developed as in the Appendix. The development of the program



(a)



(b)

Fig. 7. For example 2: (a) two-dimensional electrostatic problem, (b) solution region divided into 25 triangular elements.

basically involves four steps indicated in the program and explained as follows.

Step 1: This involves inputting the necessary data defining the problem. This is the only step that depends on the geometry of the problem at hand. Through a data file, we input the number of elements, the number of nodes, the number of fixed nodes, the prescribed values of the potentials at the fixed nodes, the x and y coordinates of all nodes, and a list identifying the nodes belonging to each element in the order of the local numbering 1-2-3. For the problem in Fig. 7, the three sets of data for coordinates, element-node relationship, and prescribed potentials at fixed nodes are shown in Tables I, II, and III, respectively. The data are inputted according to the corresponding format in the program in the Appendix.

Step 2: This step entails finding the element coefficient matrix $[C^{(e)}]$ for each element and the global matrix $[C]$. The procedure explained in the previous example is applied. The matrix $[B]$ on the right-hand side of (37) is also obtained at this stage.

TABLE I
NODAL COORDINATES OF THE FINITE ELEMENT MESH OF FIG. 7.

Node	x	y	Node	x	y
1	0.0	0.0	12	0.0	0.4
2	0.2	0.0	13	0.2	0.4
3	0.4	0.0	14	0.4	0.4
4	0.6	0.0	15	0.6	0.4
5	0.8	0.0	16	0.0	0.6
6	1.0	0.0	17	0.2	0.6
7	0.0	0.2	18	0.4	0.6
8	0.2	0.2	19	0.0	0.8
9	0.4	0.2	20	0.2	0.8
10	0.6	0.2	21	0.0	1.0
11	0.8	0.2			

TABLE II
ELEMENT-NODE IDENTIFICATION

	Local Node no.				Local Node no.		
Element	1	2	3	Element	1	2	3
1	1	2	7	14	9	10	14
2	2	8	7	15	10	15	14
3	2	3	8	16	10	11	15
4	3	9	8	17	12	13	16
5	3	4	9	18	13	17	16
6	4	10	9	19	13	14	17
7	4	5	10	20	14	18	17
8	5	11	10	21	14	15	18
9	5	6	11	22	16	17	19
10	7	8	12	23	17	20	19
11	8	13	12	24	17	18	20
12	8	9	13	25	19	20	21
13	9	14	13				

TABLE III
PRESCRIBED POTENTIALS AT FIXED NODES

Node	Prescribed Potential	Node	Prescribed Potential
1	0.0	18	100.0
2	0.0	20	100.0
3	0.0	21	50.0
4	0.0	19	0.0
5	0.0	16	0.0
6	50.0	12	0.0
11	100.0	7	0.0
15	100.0		

Step 3: The global matrix obtained in the previous step is inverted by calling subroutine INVERSE. The values of the potentials at all nodes are obtained by matrix multiplication as in (31)b. Instead of inverting the global matrix, it is also possible to solve for the potentials at the nodes using Gaussian elimination technique.

Step 4: This involves outputting the result of the computation. The output data for the problem in Fig. 7 are presented in Table IV.

The validity of the result in Table IV is checked using finite difference method. From the finite difference analysis, the potentials at the free nodes are obtained as

TABLE IV
OUTPUT DATA OF THE PROGRAM IN FIG. 7

No. of Nodes = 21
No. of Elements = 25
No. of Fixed Nodes = 15

Node	X	Y	Potential
1	0.00	0.00	0.000
2	0.20	0.00	0.000
3	0.40	0.00	0.000
4	0.60	0.00	0.000
5	0.80	0.00	0.000
6	1.00	0.00	50.000
7	0.00	0.20	0.000
8	0.20	0.20	18.182
9	0.40	0.20	36.364
10	0.60	0.20	59.091
11	0.80	0.20	100.000
12	0.00	0.40	0.000
13	0.20	0.40	36.364
14	0.40	0.40	68.182
15	0.60	0.40	100.000
16	0.00	0.60	0.000
17	0.20	0.60	59.091
18	0.40	0.60	100.000
19	0.00	0.80	0.000
20	0.20	0.80	100.000
21	0.00	1.00	50.000

$$V_8 = 15.41 \quad V_9 = 26.74 \quad V_{10} = 56.69$$

$$V_{13} = 34.88$$

$$V_{14} = 65.41 \quad V_{17} = 58.72\ V.$$

Although the result obtained using finite difference is considered more accurate in this problem, increased accuracy of finite element analysis can be obtained by dividing the solution region into greater number of triangular elements. As referred to earlier, the finite element method has a major advantage over the finite difference method. While in finite difference method, field quantities are obtained only at discrete positions in the solution region, they can be obtained at any point in the solution region in finite element method.

CONCLUSIONS

The finite element method was originally expounded for civil engineering applications but has recently seen increasing applications in EM. A brief, tutorial introduction to the method has been presented. The method is applied specifically to Laplacian electrostatic field problems. Suggestions are offered on how the concepts developed can be extended to other types of EM problems.

It should be noted that one problem associated with the finite element method is inputting the data defining the geometry of the solution region. Since a large amount of data is involved, a mistake can easily be made in entering this data. This problem can be attacked in three different ways. One is by having an in-built subroutine for data verification. Another method is using computer graphics to plot the mesh before analyzing the data. These two methods enable recognition of mistakes. A third method is using an automatic mesh generating scheme, e.g., [11], [12]; this labor-saving method of discretizing the solution region into triangular elements eliminates the problem. Another major problem in finite element analysis is the relatively large amount of computer memory and time required. With the advent of batch processing and time-sharing, the problem has been alleviated to some degree.

In spite of the limitations and problems, the finite element method has successfully been applied to various EM problems. For such applications, the reader is referred to the literature, e.g., [4], [7].

APPENDIX: COMPUTER PROGRAM FOR EXAMPLE 2

```
C     FINITE ELEMENT SOLUTION OF LAPLACE'S EQUATION FOR
C     TWO-DIMENSIONAL PROBLEMS
C     TRIANGULAR ELEMENTS ARE USED
C
C     ND = NO. OF NODES
C     NE = NO. OF ELEMENTS
C     NP = NO. OF FIXED NODES (WHERE POTENTIAL IS PRESCRIBED)
C     NDP(I) = NODE NO. OF PRESCRIBED POTENTIAL, I = 1,2,...NP
C     VAL(I)) = VALUE OF PRESCRIBED POTENTIAL AT NODE NDP(I)
C     NL(I,J) = LIST OF NODES FOR EACH ELEMENT I, WHERE
C     J = 1, 2, 3  IS THE LOCAL NODE NUMBER
C     CE(I,J) = ELEMENT COEFFICIENT MATRIX
C     C(I,J)  = GLOBAL COEFFICIENT MATRIX
C     B(I) = RIGHT-HAND SIDE MATRIX IN THE SYSTEM OF
C     SIMULTANEOUS EQUATIONS; SEE EQ.(31) OR EQ.(37)
C     X(I), Y(I) = GLOBAL COORDINATES OF NODE I
C     XL(J), YL(J) = LOCAL COORDINATES OF NODE J = 1,2,3
C     V(I) = POTENTIAL AT NODE I
C     MATRICES P(I) AND Q(I) ARE DEFINED IN EQ.(32)

      DIMENSION  X(100), Y(100), C(100,100), CE(100,100)
      DIMENSION  B(100), NL(100,3), NDP(100), VAL(100)
      DIMENSION  V(100), P(3), Q(3), XL(3), YL(3)

C     ********************************************************
C     FIRST STEP - INPUT DATA DEFINING GEOMETRY AND
C                  BOUNDARY CONDITIONS
C     ********************************************************
      READ(5,10) NE,ND, NP
10    FORMAT(3I3)
      READ(5,20)( I, ( NL(I,J), J=1,3),I=1,NE)
20    FORMAT(4I3)
      READ(5,30) (I, X(I), Y(I), I=1,ND)
30    FORMAT(I3, 2F6.2)
      READ(5,40) ( NDP(I), VAL(I), I=1,NP)
40    FORMAT(I3,F6.2)
C     ********************************************************
C     SECOND STEP - EVALUATE COEFFICIENT MATRIX FOR EACH
C                   ELEMENT AND ASSEMBLE GLOBALLY
C     ********************************************************
      DO 50 M =1, ND
      B(M)=0.0
      DO 50 N=1,ND
      C(M,N) = 0.0
50    CONTINUE
      DO 140  I = 1, NE
C     FIND LOCAL COORDINATES XL(J), YL(J) FOR ELEMENT I
      DO 60 J=1,3
      K=NL(I,J)
      XL(J) = X(K)
      YL(J) = Y(K)
60    CONTINUE
      P(1) = YL(2) - YL(3)
      P(2) = YL(3) - YL(1)
      P(3) = YL(1) - YL(2)
      Q(1) = XL(3) - XL(2)
      Q(2) = XL(1) - XL(3)
      Q(3) = XL(2) - XL(1)
      AREA = 0.5*ABS( P(2)*Q(3) - Q(2)*P(3) )
C     DETERMINE COEFFICIENT MATRIX FOR ELEMENT I
      DO 70 M=1,3
      DO 70 N=1,3
      CE(M,N) = ( P(M)*P(N) + Q(M)*Q(N) )/(4.0*AREA)
70    CONTINUE
C     ASSEMBLE GLOBALLY - FIND C(I,J) AND B(I)
      DO 130 J=1,3
      IR=NL(I,J)
C     CHECK IF ROW CORRESPONDS TO FIXED NODE
      DO 80 K=1,NP
      IF(IR.EQ.NDP(K)) GO TO 120
80    CONTINUE
      DO 110 L=1,3
      IC = NL(I,L)
C     CHECK IF COLUMN CORRESPONDS TO FIXED NODE
      DO 90 K=1,NP
      IF( IC.EQ.NDP(K) ) GO TO 100
90    CONTINUE
      C(IR,IC) = C(IR,IC)  +  CE(J,L)
      GO TO 110
100   B(IR) = B(IR) - CE(J,L)*VAL(K)
110   CONTINUE
      GO TO 130
```

```
120   CONTINUE
      C(IR,IR) = 1.0
      B(IR) = VAL(K)
130   CONTINUE
140   CONTINUE
C     ***********************************************
C     THIRD STEP - SOLVE THE RESULTING SYSTEM OF
C                  SIMULTANEOUS EQUATIONS
C     ***********************************************
      NMAX=100
      CALL INVERSE(C,ND,NMAX)
      DO 150  I=1,ND
      V(I) = 0.0
      DO 150 J=1,ND
      V(I) = V(I) + C(I,J)*B(J)
150   CONTINUE
C     ***********************************************
C     FOURTH STEP - OUTPUT THE RESULTS, THE POTENTIAL
C                   V(I) AT NODE I,  I = 1,2,....ND
C     ***********************************************
      WRITE(6,160) ND,NE,NP
160   FORMAT(2X,'NO. OF NODES = ',I3,2X,'NO. OF ELEMENTS =',
     1   I3,2X,'NO. OF FIXED NODES = ',I3,/)
      WRITE(6,170)
170   FORMAT(2X,'NODE',5X,'X        Y',7X,'POTENTIAL',/)
      WRITE(6,180) (I, X(I), Y(I), V(I), I=1,ND )
180   FORMAT(2X,I3,2X,F6.2,2X,F6.2,2X,F10.4,/)
      STOP
      END



C     SUBROUTINE INVERSE
C     SX IS THE MATRIX TO BE INVERTED; IT IS DESTROYED
C     IN THE COMPUTATION AND REPLACED BY THE INVERSE
C     N IS THE ORDER OF SX
C     IDM IS THE DIMENSION OF SX

      SUBROUTINE INVERS(SX,N,IDM)
      DIMENSION SX(IDM,IDM)

      ESP=1.0E-5
      DO 50 K=1,N
      DO 30 J=1,N
      IF(J.EQ.K) GO TO 30
      IF( ABS(SX(K,K)) ) 20,10,20
10    SX(K,K)=ESP
20    SX(K,J)=SX(K,J)/SX(K,K)
30    CONTINUE
      SX(K,K)=1.0/SX(K,K)
      DO 40 I=1,N
      IF(I.EQ.K) GO TO 40
      DO 40 J=1,N
      IF(J.EQ.K) GO TO 40
      SX(I,J)=SX(I,J) - SX(K,J)*SX(I,K)
40    CONTINUE
      DO 50 I=1,N
      IF(I.EQ.K) GO TO 50
      SX(I,K)= - SX(I,K)*SX(K,K)
50    CONTINUE
      RETURN
      END
```

ACKNOWLEDGMENT

The author would like to express his appreciation to Y. Kim and L. Agba for their review and helpful comments on the manuscripts. The corrections and comments of the reviewers are also appreciated.

REFERENCES

[1] R. Courant, "Variational methods for the solution of Problems of Equilibrium and vibrations," *Bull. Am. Math. Soc.*, vol. 49, 1943.

[2] J. F. Hoburg and J. L. Davis, "A student-oriented finite element program for electrostatic potential problems," *IEEE Trans. Educ.*, vol. E-26, pp. 138-142, Nov. 1983.

[3] ——, "Enhanced capabilities for a student-oriented finite element electrostatic potential program," *IEEE Trans. Educ.*, vol. E-28, pp. 25-28, Feb. 1985.

[4] S. R. H. Hoole and P. R. P. Hoole, "Finite element programs for teaching electromagnetics," *IEEE Trans. Educ.*, vol. E-29, Feb. 1986.

[5] R. F. Harrington, *Field Computation by Moment Methods*. Malabar: Kriger, 1968.

[6] W. H. Hayt, *Engineering Electromagnetics*. New York: McGraw-Hill, 1981, pp. 121, 184-192.

[7] P. P. Silvester and R. L. Ferrari, *Finite Elements for electrical Engineers*. Cambridge: Cambridge Univ. Press, 1983, pp. 1-32.

[8] C. R. Paul and S. A. Nasar, *Introduction to Electromagnetic Fields*. New York: McGraw-Hill, 1982, pp. 465-472.

[9] O. W. Andersen, "Laplacian electrostatic field calculations by finite elements with automatic grid generation," *IEEE Trans. Power Apparatus Syst.*, vol. PAS-92, no. 5, Sept./Oct., pp. 1485-1492, 1973.

[10] D. R. J. Owen and E. Hinton, *A Simple Guide to Finite Elements*. Swansea: Pineridge Press, 1980, pp. 47-79.

[11] E. Hinton and D. R. J. Owen, *An Introduction to Finite Element Computations*. Swansea: Pineridge Press, 1985, pp. 328-326.

[12] W. R. Buell and B. A. Bush, "Mesh generation—A survey," *Trans. ASME, J. Engng. Ind.*, pp. 332-338, Feb. 1973.

Matthew N. O. Sadiku graduated from Ahmadu Bello University, Zaria (Nigeria) and received the M.S. and Ph.D. degrees from Tennessee Tech. University, Cookeville.

He is currently an Assistant Professor of Electrical Engineering at Temple University, Philadelphia. PA. He was formerly at Florida Atlantic University, Boca Raton. He is the author of *Elements of Electromagnetics* (HRW, 1989). His current research interests are in numerical techniques in electromagnetics, local area computer networks, and superconductivity.